
Abstract. Memcomputing logic gates generalize the tra-
ditional Boolean logic gates for operation in the reverse di-
rection. According to the literature, this functionality en-
ables efficient solution of computationally intensive prob-
lems, including factorization and NP-complete problems. To
approach the deployment of memcomputing gates in hard-
ware, this paper introduces SPICE models of memcomputing
logic gates following their original definition. Using these
models, we demonstrate the behavior of single gates as well
as small self-organizing circuits. We have also corrected
some inconsistencies in the prior literature. Notably, the
correct schematics of the dynamic correction module is re-
ported here for the first time. Our work makes memcom-
puting more accessible to those interested in this emerging
computing technology.

1. Introduction
Digital memcomputers are an emerging class of un-

conventional computing systems developed to efficiently
solve factorization and combinatorial optimization prob-
lems [1], [2]. Fundamentally, these are complex dynami-
cal systems with deterministic continuous dynamics whose
phase space contains a fixed point attractor corresponding to
the problem solution (or multiple attractors if several solu-
tions are possible). According to Traversa and Di Ventra,
when implemented in hardware, digital memcomputing ma-
chines offer a polynomial-time solution to factorization and
NP-complete problems [1]. Moreover, it has been argued
that the dynamics of digital memcomputing machines (with
solutions) is deterministic, non-chaotic, without periodic or-
bits [3], [4], and topologically robust against perturbations
and noise [5], [6]. If a solution exists, it is certain that it
will be discovered. Otherwise, there is no equilibrium. In
such cases, some or all of the gates will continually change in
an effort to self-organize into the logically consistent state of
the circuit. However, such a state does not exist in problems
without a solution.

So far, the research on digital memcomputing has been
substantially focused on software simulations of ordinary dif-
ferential equations representing the circuit dynamics. A re-
cent benchmark [7] indicates that the memcomputing solu-
tion for a specific class of difficult problems is characterized
by an exponent similar to that found for other solvers (such
as Toshiba simulated bifurcation machines [8] and Fujitsu
digital annealers [9]). Further progress could be made if
computing was implemented on the hardware. For more in-
formation on competitive computing approaches, we refer
the interested reader to [7] and the references therein.

Three designs of memcomputing logic gates are avail-
able in the literature. The most complex is the original de-
sign [1] (Design I) that is presented in Fig. 1. According to
Fig. 1, self-organizing AND, OR or XOR can be built using
12 memristive elements, 15 voltage-controlled voltage gen-
erators, and several resistors. Moreover, some auxiliary cir-
cuitry is required to ensure that the final states of these gates
(operating in the continuous or analog domain) are binary.
Bearden et al. [10] introduced a simplified design of self-
organizing AND and OR (Design II) [10] in which each gate
requires 5 memristive elements, 6 voltage generators and few
resistors. In principle, the simplified gates can perform the
same tasks as the original ones. A study shows that Design II
gates can be built using physical memristive devices [14]. In
the third design (Design III), the gates are defined by a set of
differential equations [15]. In fact, Design III is a variation
of analogSAT [16], [17]. Recently, Design III calculations
were implemented on an FPGA board [18].

The purpose of this work is to develop SPICE models
of Design I self-organizing logic gates [1]. During the last
decade or so, the SPICE modeling of adaptive circuit ele-
ments (known as memrisitive, memcapacitive, and memin-
ductive systems [11,12,19]) has become increasingly impor-
tant and resulted in various SPICE models of deterministic
memelements (see, for instance, [20–28]). A notable recent
development is the simulation of probabilistic memristive
devices in SPICE [29]. SPICE is a general-purpose circuit
simulation program [30], [31]. In SPICE, the circuit can be
built graphically and then numerically simulated using pre-
defined or user-defined models of individual circuit compo-
nents. Such user-defined models of Design I self-organizing
logic gates [1] are formulated in this paper.

Fig. 1. Left panel: the universal self-organizing logic gate is composed of three dynamic correction modules (DCMs) and two resistors. Right
panel: internal structure of the dynamic correction module (incorrect). Here, the resistive memories 𝑀 [11–13] have minimum and

maximum resistances 𝑅on and 𝑅off, respectively, and the resistance of the resistors is 𝑅 = 𝑅off. 𝐿𝑀𝑗 -s and 𝐿𝑅 are voltage-controlled

voltage generators. As we explain in the text, in the right panel, the polarity of all memristive elements must be reversed. Reprinted

from [1], with the permission of AIP Publishing.

In this work, we used LTSpice XVII (Analog Devices)

as a simulation tool. Our SPICE models may need minor

adjustments to be used on other SPICE simulators (e.g.,

PSPICE, Ngspice).

The paper is organized as follows. We start with the

introduction of Design I self-organizing logic gates (Sec. 2)

that is followed by a presentation of their SPICE models

(Sec. 3). In Sec. 4, we give examples of SPICE simulations

of individual gates and circuits thereof. Section 5 concludes

the paper. Complete listings of LTspice codes are given in

Appendix C.

2. Self-Organizing Gates and Circuits
Self-organizing logic gates generalize traditional logic

gates for operation in the reverse direction [1], [2]. In these

gates, each terminal serves the double function of input and

output. Although self-organizing gates operate in analog

mode, the auxiliary circuitry (voltage-controlled differential

current generators presented below) and the gate design en-

sure that the final states are binary. In what follows, Boolean

1 and 0 are represented by 𝑣c = 1V and −𝑣c = −1V voltage
levels, respectively.

The logic behind the construction of self-organizing

gates can be partially captured from the following excerpt

from [1]: “if the gate is connected to a network and the gate

configuration is correct, no current flows from any terminal

(the gate is in stable equilibrium). Otherwise, a current of

the order of 𝑣c/𝑅on flows with the sign opposite to the sign

of the voltage at the terminal.” Below, this property of self-

organizing gates is used to demonstrate that the polarity of

the memristive elements in Fig. 1 circuit must be reversed for

the correct gate operation.

Transformation of traditional Boolean logic circuits into

Design I self-organizing logic circuits [1] involves the follow-

ing steps:

• Replacing the traditional logic gates with self-

organizing gates of the same type.

• Representing the external input signals by constant-
value voltage sources.

• Adding auxiliary circuitry: A voltage-controlled dif-
ferential current generator (VCDCG) is added to each

node, but not to the nodes that are used for input signals.

By node, we mean either the point of connection of two

or more gate terminals or an unconnected gate terminal.

• External input signals are applied at the initial moment
of time and stay constant over time. The end of dynam-

ics indicates that a solution is found and can be read.

The infinite dynamics implies the absence of a solution.

For the sake of completeness, we next provide the mini-

mal description of the circuit components in the Design I

self-organizing logic circuits that is required for their im-

plementation in SPICE. These definitions were extracted

from [1,2, 32].

2.1 Voltage-Controlled Voltage Generators
Consider the structure of the universal self-organizing

gate shown in Fig. 1. Its ultimate functionality (e.g., self-
organizing AND, OR, or XOR) is defined by the equations
that govern voltage-controlled voltage generators (VCVGs)
𝐿𝑀1 − 𝐿𝑀4 and 𝐿𝑅. According to [1], the voltage across
VCVG is a linear function of the gate voltages

𝑣VCVG = 𝑎1𝑣1 + 𝑎2𝑣2 + 𝑎0𝑣0 + 𝑑𝑐 (1)

where 𝑣1, 𝑣2, and 𝑣0 are the gate voltages, and 𝑎1, 𝑎2, 𝑎0,
and 𝑑𝑐 are the constants. For convenience, these constants
are specified in Tab. A1 (Appendix A).

2.2 Memristive Elements
Having defined the voltage-controlled voltage genera-

tors, next we consider the memristive elements 𝑀 in Fig. 1.
The response of the memristive system 𝑗 is described by
Ohm’s law

𝑣𝑀 𝑗
(𝑡) = 𝑀 (𝑥 𝑗 )𝑖𝑀 𝑗

(𝑡) (2)

where 𝑣𝑀 𝑗
is the voltage (defined with respect to the

thick-bar terminal of the circuit symbol of 𝑀), 𝑖𝑀 𝑗
is the

current,

𝑀 (𝑥 𝑗 ) = (𝑅off − 𝑅on) 𝑥 𝑗 + 𝑅on (3)

is the state-dependent resistance (memristance), 𝑥 𝑗 ∈ [0, 1]
is the internal state variable [11], 𝑅on and 𝑅off are the on- and
off-state resistances.

The dynamics of 𝑥 𝑗 follows the ordinary differential
equation:

d𝑥 𝑗

d𝑡
= −𝛼ℎ(𝑥 𝑗 , 𝑣𝑀 𝑗

)
[
(𝑅off − 𝑅on) 𝑥 𝑗 + 𝑅on

]−1
𝑣𝑀 𝑗

(4)

where 𝛼 is the constant and the function ℎ(𝑥, 𝑣𝑀 ) influences
the dynamics of the internal state. While many choices for
ℎ(𝑥, 𝑣𝑀 ) are available, following [1], we use

ℎ(𝑥, 𝑣𝑀 ) = 𝜃 (𝑥) 𝜃 (𝑣𝑀 ) + 𝜃 (1 − 𝑥) 𝜃 (−𝑣𝑀 ) (5)

where 𝜃 (. . . ) is the unit step function. According to (4)
and (5), 𝑥 𝑗 decreases down to 𝑥 𝑗 = 0 at positive voltages. At
negative voltages, 𝑥 𝑗 increases to 𝑥 𝑗 = 1. It is not difficult
to recognize that (2), (4), and (5) correspond to the ideal
memristor model [33]. The limitations of this model are well
known [19]. Note that Table A2 (Appendix A) lists the values
of parameters used in the prior simulations [1]. Equation (5)
corresponds to 𝑉𝑡 = 0 and 𝑘 = ∞. For definition of 𝑉𝑡 and 𝑘 ,
see [1].

Moreover, a parasitic device capacitance is taken into
account using a constant-value capacitor of capacitance 𝐶

connected in parallel to every memristive element. These
capacitors are not shown in Fig. 1 explicitly but taken into
account in the model.

2.3 Voltage-Controlled Differential Current
Generators
Finally, we introduce equations describing voltage-

controlled differential current generators. These generators
are second-order dynamical systems whose evolution fol-
lows [1], [32]:

d𝑖DCG, 𝑗

d𝑡
= 𝜃

(
𝑠 𝑗 −

1
2

)
𝑓DCG (𝑣DCG, 𝑗 ) −

𝛾𝜃

(
1
2
− 𝑠 𝑗

)
𝑖DCG, 𝑗 , (6)

d𝑠 𝑗
d𝑡

= 𝑓𝑠
(
iDCG, 𝑠 𝑗

)
. (7)

Here, 𝛾 is the constant, 𝑣DCG, 𝑗 is the voltage at the node that
VCDCG is connected to, 𝑖DCG, 𝑗 is the current, iDCG is the
vector of the currents of all VCDCGs, and 𝑠 𝑗 is the second
state variable. Note that the above equation corresponds to
𝛿𝑠 = 0 in Tab. A2 (Appendix A). For the definition of 𝛿𝑠,
see [1]. Our specific realization of 𝑓DCG (𝑥) in (6) is based
on arc tangent functions (one of the suggested realizations of
𝑓DCG (𝑥) [1]). Specifically, we use

𝑓DCG (𝑥) =
2𝑞
𝜋

(
arctan

[
𝑚1𝜋

2𝑞
(𝑥 + 𝑣c)

]
+

arctan
[
𝑚0𝜋

2𝑞
𝑥

]
+ arctan

[
𝑚1𝜋

2𝑞
(𝑥 − 𝑣c)

] )
(8)

where 𝑞, 𝑚0, 𝑚1, and 𝑣c are the constants. Equation (8) is
illustrated in Fig. 2(a).

In (7), the function 𝑓𝑠
(
iDCG, 𝑠 𝑗

)
is defined as

𝑓𝑠
(
iDCG, 𝑠 𝑗

)
= −𝑘𝑠𝑠 𝑗 (𝑠 𝑗 − 1) (2𝑠 𝑗 − 1)−

𝑘𝑖

(
1 −

∏
𝑝

𝜃

(
𝑖2min − 𝑖2DCG, 𝑝

)
−

∏
𝑝

𝜃

(
𝑖2max − 𝑖2DCG, 𝑝

))
. (9)

Here, 𝑘𝑠 , 𝑘𝑖 , 𝑖min, and 𝑖max are the positive constants
(𝑖min < 𝑖max). Equation (9) corresponds to 𝛿𝑖 = 0 in Tab. A2
(Appendix A). For the definition of 𝛿𝑖, see [1].

Importantly, unlike [1], [32] we use the minus sign in
front of the 𝑘𝑖 term in (9). The minus sign is required to im-
plement the following anticipated purpose of 𝑠 𝑗 -s: the reset
of all |𝑖DCG, 𝑗 | to below 𝑖min as soon as at least one of |𝑖DCG, 𝑗 |
exceeds 𝑖max.

Equation (9) is illustrated in Fig. 2(b). Figure 2(b) (left
panel) shows that when the absolute value of all currents is
less than 𝑖min, the function 𝑓𝑠

(
iDCG, 𝑠 𝑗

)
has a single zero at

𝑠 > 1, which is a stable fixed point of (7). When the ab-
solute value of at least one 𝑖DCG, 𝑝 is larger than 𝑖max, the
stable fixed point is located at 𝑠 < 0, see Fig. 2(c) (right
panel). In the intermediate case, there are two stable fixed
points and one unstable fixed point (Fig. 2(b) (middle panel)).
Therefore, when the absolute value of one of the currents ex-
ceeds 𝑖max, all variables 𝑠 𝑗 – which are described by identical
equations – start drifting toward the negative stable fixed
point, causing the relaxation of VCDCG currents (through
the last term in (6)). The normal response of VCDCGs (due
to the first term ino. (6)) is restored later, after the condition
|𝑖DCG, 𝑗 | < 𝑖min for all 𝑗 is satisfied.

Fig. 2. Functions (a) 𝑓DCG (𝑥 ) and (b) 𝑓𝑠
(
iDCG, 𝑠 𝑗

)
defined

by (8) and (9), respectively. These graphs were obtained
using the parameters values from Tab. A2 (Appendix A).

3. SPICE Models

3.1 Basic Details
Our SPICE models are based on the parameters’ val-

ues from [1] (Tab. A2, Appendix A) with the assumption
that these values are given in the SI units. An issue is that
the small resistances lead to large currents (of the order of
amperes) and some other parameters lead to slow dynamics
(on the scale of seconds). In particular, the on-and off-state
resistances, 𝑅on = 0.01Ω and 𝑅off = 1Ω, are much smaller
than the on- and off-state resistances in experimental devices
that typically range from kiloohms to megaohms. In our LT-
spice models, we have introduced two scaling factors, 𝑧𝐼 and
𝑧𝑡 . These factors are used to rescale the currents and time
in typical ranges for electronics. For consistency, the same
values of 𝑧𝐼 and 𝑧𝑡 must be utilized in all LTspice models
(Appendix C). The results reported here were obtained using
𝑧𝐼 = 105 and 𝑧𝑡 = 103.

Appendix C contains LTspice models of self-organizing
AND, OR, and XOR, memristive elements, and voltage-
controlled differential current generators (Listings C1–C5).
These models were formulated following the common prac-
tices in SPICE modeling [22]. For example, to inte-
grate the differential equations (4), (6), and (7), we use
capacitors that are charged or discharged with voltage-
controlled current sources representing the right-hand sides
of these equations, etc.

Figure 3 shows the current-voltage curves for the mem-
ristive element. To obtain these curves, we used the SPICE
model from Listing C4. According to Fig. 3, the memris-
tance decreases at 𝑉𝑀 > 0 and increases at 𝑉𝑀 < 0. The
frequency dependence of the current-voltage curves in Fig. 3
is typical for memristive systems [11], [12].

Fig. 3. Current-voltage curves of the memristive element sub-
jected to a sinusoidal voltage. These curves were ob-
tained using the SPICE model in Lst. C4 (Appendix C).

We have found that the polarity of the memristive el-
ements in Fig. 1 is incorrect. To understand this, one can
evaluate the total terminal current in a gate terminal assum-
ing a logically consistent state. As we have mentioned above,
in this case, the total terminal current must be zero (see
the second paragraph in Sec. 2.1). For example, a simple
calculation shows that the current at terminal 1 of AND at
𝑉1 = 𝑉2 = 𝑉𝑜 = −1 V is 𝐼1 = 2/𝑀1 − 2/𝑅off and the voltage
across 𝑀1 in Fig. 1 circuit is +2 V. As positive voltages drive
𝑀1 into 𝑅on, 𝐼1 = 2/𝑅on−2/𝑅off > 0. Thus, to satisfy 𝐼1 = 0,
the polarity of 𝑀1 must be reversed.

As all variables 𝑠 𝑗 are described by the same Equa-
tion (7), we represent these variables by a single variable
𝑠 ≡ 𝑠 𝑗 for all 𝑗 . In SPICE, we have defined an s-block as
the component that integrates (7) (see the second model in
Listing C5 and Fig. B1 in Appendix B). The s-block has 8
voltage inputs for the voltage signals encoding 𝑖DCD, 𝑗 -s and
a single output, which is 𝑠. The output of s-block must
be connected to all VCDCGs (to terminals 4). The inputs
of the s-block are taken from terminals 3 of the VCDCGs.
The SPICE model of the s-block in Listing C5 can be directly
used with up to 8 VCDCGs in the circuit and can be easily
expanded to a larger number of VCDCGs.

3.2 Important Implementation Notes

1. The correct version of Design I self-organizing gates
involves two resistors (Fig. 1, left). In [2], it is shown
without resistors. In [32], it is presented with memris-
tors instead of resistors.

2. The polarity of memristors in Fig. 1 has been reversed,
see Fig. 4.

3. We emphasize that unlike [1], [32], we use the minus
sign in front of 𝑘𝑖 in (9).

4. The values 𝑘𝑖 = 10−7 and 𝑘𝑠 = 10−7 from [1], [32] are
too small. To enable the correct operation of the reset
feature in VCDCGs, we use 𝑘𝑖 = 𝑘𝑠 = 2E3.

5. Memristive elements subjected to zero voltage and as-
sociated voltage sources have not been included in the
models of AND and OR (e.g., 𝐿𝑀2 and 𝐿𝑀4 and asso-
ciated memristive elements in the DCM of terminal 1
of AND).

6. All 𝑠 𝑗 -s are implemented using a single variable 𝑠.

7. To suppress high voltage spikes1 and improve the con-
vergence, we have increased 𝑅on to 0.05, decreased 𝑖max
to 10, and added a capacitor in each VCDCG (C1 in
Listing C5).

8. To enable the random initial states of memristive el-
ements, the option "Use the clock to reseed the MC
generator" must be checked in LTSpice XVII. The tran-
sient analysis was performed using the option uic.

9. For reproducibility of our results, the initial states of
the memristive elements are chosen from a flat random
distribution between 0.18 and 0.22. All other initial
values are selected deterministically.

4. Simulation Examples

4.1 Single Gates
This subsection exemplifies the behavior of self-

organizing gates using the OR gate as an example. First,
we consider the traditional (direct) operation, wherein the
voltage signals are applied to the traditional inputs of the
gate. Second, we explore the reverse operation (not available
with the usual OR).

LR

LM1

LM3

M1

M4

M2

M3

LM2

V

LM4

R1

Memristors and More\Memcomputing DMM SPICE___________________________________\final\DCM\Draft1.

Fig. 4. Correct schematics of the dynamic correction module.

1In the circuit based on original parameters, spikes can be of quite extreme magnitude (e.g., several hundred thousand volts). These spikes have been
associated with instantons, see [2].



RADIOENGINEERING, VOL. 32, NO. 4, DECEMBER 2023 547

Figure 5(a) shows the simulated circuit for the direct
operation of the self-organizing OR. In this circuit, the input
signals are applied using pulsed voltage sources connected
to terminals 1 and 2 of U1. The output terminal of U1 (not
connected to any voltage source) is driven by a VCDCG (see
the circuit transformation rules in Sec. 2). Figure 5(b) shows
the gate response. Clearly, except for short transients, the
gate reproduces the truth table of OR. In Fig. 5(b), the curve
𝑉VCDCG represents the current in VCDCG U2 (without ac-
counting for the current of C1). We note that this current
fluctuates at about zero.

Next, consider the reverse operation of the self-
organizing OR. Figure 6(a) and (b) show two slightly dif-
ferent circuits used in our simulations. The difference is
that in Fig. 6(a) we use two terminals (terminals 2 and 𝑜)
as input and one terminal (1) as output, while in Fig. 6(b)
there is a single input terminal (𝑜) and two output terminals
(1 and 2). As before, we use VCDCGs to ensure binary states
at the output terminals.

Figure 6(c) shows the gate voltages for the circuit in
Fig. 6(a). In this presented realization of circuit dynamics,
after a transient, the voltage at the first terminal, 𝑉1 (𝑡), be-
comes equal to −1 V. When 𝑉𝑜 = 1 V, the truth table of OR
is satisfied. In the opposite case, the gate shows a reasonable
behavior as it chooses 𝑉1 = −1 V over 𝑉1 = 1 V determin-
istically. In some other runs, instead of 𝑉1 (𝑡) = −1 V, the
voltage at terminal 1, after a short transient, repeated the ap-
plied voltage 𝑉𝑜 (𝑡). In this case, again, the truth table of OR
is satisfied whenever 𝑉𝑜 (𝑡) = 1 V.

Finally, consider the response of the self-organizing
OR in Fig. 6(b). Using a flat random distribution of ini-
tial states of memristive elements from 0 to 1, in each run
we observed one of the following general responses: (i)
𝑉1 (𝑡) = 𝑉2 (𝑡) = 𝑉𝑜 (𝑡) (as in Fig. 6(d)), 𝑉1 (𝑡) = −1 V,
𝑉2 (𝑡) = 𝑉𝑜 (𝑡), or (iii) 𝑉1 (𝑡) = 𝑉𝑜 (𝑡), 𝑉2 (𝑡) = −1 V. Clearly,
all of these cases are consistent with the truth table of the OR.

Overall, we conclude that the self-organizing OR cor-
rectly reproduces the truth table of OR (whenever possible)
regardless of the role of each terminal as input or output.
The response may be different in different runs (but always
correct). We have verified that the same is true for self-
organizing AND and XOR.

4.2 Circuits of Self-Organizing Gates
As an example of a circuit of self-organizing gates,

consider a self-organizing 2-bit by 2-bit multiplier. Its
schematics is presented in Fig. 7(a). The self-organizing
multiplier involves eight self-organizing gates and eight
voltage-controlled differential current generators. This cir-
cuit was designed based on the conventional 2-bit by 2-
bit binary multiplier using the circuit transformation rules
outlined in Sec. 2.

The circuit in Fig. 7(a) uses four constant voltage
sources V3–V6 to encode the number to factorize (it is 6
in Fig. 7(a)). We emphasize that the signals P0–P3 serve as
the input. The output signals are A0, A1, B0, and B1, which
are the bits of two factors (A0 and B0 are the least signifi-
cant bits). These factors are found through the deterministic
dynamics of the self-organizing multiplier.

Examples of circuit dynamics are demonstrated in
Figs. 7(b)–(f). In particular, Figs. 7(b) and (c) show that
the result may be different in different runs. Specifically,
these graphs indicate that number 2 can be presented as 2 · 1
or 1 · 2. This ability to identify different solutions is related
to the random choice of initial states of memristive elements.
Two distinct solutions are observed when two sets of initial
states belong to different basins of attraction.

Figures 7(e) and (f) indicate that the factorization of
some numbers can be more difficult than others (using some
close initial conditions). In our simulations of the self-
organizing multiplier, the most difficult was the factorization
of 1. In this case, most frequently, we have observed the
transition to a limit cycle behavior as the one in Fig. 7(f)
and quite occasionally the correct solution to the problem
(1 = 1 · 1).

We have verified that the existence of the limit cycle
(Fig. 7(f)) is not related to certain modifications to the pa-
rameters that we made. In particular, the limit cycle was
observed in the circuit without C1 (in the VCDCG model)
and with prior values of 𝑅on, 𝑞, and 𝑖max (from Tab. A2).
In a longer simulation, it was observed that the limit cycle
continues up to 100 s. To ensure that the limit cycle is not
a numerical artifact, we performed some additional simula-
tions. The use of other numerical integration methods in
LTspice (trapezoid and modified trap in addition to Gear)2,
variation of tolerances, noise addition, and the use of PSPICE
result in the same limit cycle behavior.

The existence of the limit cycle in the numerical dynam-
ics seems to contradict the statement "if the Boolean problem
the DMMs are designed to solve has a solution, the system
will always find it, irrespective of the initial conditions" in [4]
(see also [1]) for the continuous dynamics. Currently, the ex-
act reason for this is not known, and its determination is
beyond the scope of this work.

5. Conclusion
Having identified and corrected some inconsisten-

cies in the prior literature [1, 2, 32] (see Items 1–4 in
Sec. 3.2), we have formulated SPICE models of the De-
sign I self-organizing logic gates. The operation of indi-
vidual self-organizing gates and small circuits thereof has
been demonstrated.

2It is known that the basin of attraction is influenced by the discretization [34].

Fig. 5. Direct operation of self-organizing OR. (a) Circuit used in the simulations. Here, V1 and V2 are the pulsed voltage sources, U1 is the
self-organizing OR (for the LTspice code, see Listing C3), U2 is the VCDCG (for the LTspice code, see Listing C5), and U18 is the s-block
(see Listing C5 and Fig. B1 in Appendix B). The s-block implements (7) and (9). (b) Transient voltage signals at the terminals of the
self-organizing OR and VCDCG voltage output (terminal 3 of U2). The curves were displaced for the sake of clarity.

Fig. 6. Reverse operation of self-organizing OR. (a), (b) Circuits used in simulations. Here, V4 and V14 are pulsed voltage sources and V1 is
the 1 V constant voltage source. (c) Example of voltage transient signals at the terminals of the self-organizing OR in (a). (d) Example of
transient voltage signals at the terminals of the self-organizing OR in (b).

Fig. 7. Solving integer factorization problem with a self-organizing 2-bit by 2-bit multiplier. (a) Circuit used in the simulations. Here, the number
to factorize is represented by (P3,P2,P1,P0), and the factors – by (A1,A0) and (B1,B0). (b)–(f) Examples of the transient dynamics of the
self-organizing 2-bit by 2-bit multiplier. Curve 𝑠 in (f) was shifted down by 2 V for clarity.



550 Y. V. PERSHIN, SPICE MODELING OF MEMCOMPUTING LOGIC GATES

We emphasize that in future studies of these gates, spe-
cial attention should be paid among others to:

• Polarity of memristive devices in DCMs.

• Use of resistors in the schematics of the universal gate.

• Sign of the 𝑘𝑖 term in the function 𝑓𝑠
(
iDCG, 𝑠 𝑗

)
.

• Values for parameters 𝑘𝑖 and 𝑘𝑠 .

In summary, self-organizing logic gates are an interest-
ing generalization of the traditional Boolean logic gates. The
SPICE models reported in this paper offer an easy and pretty
reliable way to explore self-organization in memcomput-
ing circuits. Experimental demonstration of self-organizing
gates is an interesting project for the future.

Appendix A: Parameters

Terminal 1 Terminal 2 Out Terminal
𝑎1 𝑎2 𝑎0 𝑑𝑐 𝑎1 𝑎2 𝑎0 𝑑𝑐 𝑎1 𝑎2 𝑎0 𝑑𝑐

SO AND
𝐿𝑀1 0 –1 1 𝑣c –1 0 1 𝑣c 1 0 0 0
𝐿𝑀2 1 0 0 0 0 1 0 0 0 1 0 0
𝐿𝑀3 0 0 1 0 0 0 1 0 0 0 1 0
𝐿𝑀4 1 0 0 0 0 1 0 0 2 2 –1 −2𝑣c
𝐿𝑅 4 1 –3 −𝑣c 1 4 –3 −𝑣c –4 –4 7 2𝑣c
SO OR
𝐿𝑀1 0 0 1 0 0 0 1 0 0 0 1 0
𝐿𝑀2 1 0 0 0 0 1 0 0 2 2 –1 2𝑣c
𝐿𝑀3 0 –1 1 −𝑣c –1 0 1 −𝑣c 1 0 0 0
𝐿𝑀4 1 0 0 0 0 1 0 0 0 1 0 0
𝐿𝑅 4 1 –3 𝑣c 1 4 –3 𝑣c –4 –4 7 −2𝑣c
SO XOR
𝐿𝑀1 0 –1 –1 𝑣c –1 0 –1 𝑣c –1 –1 0 𝑣c
𝐿𝑀2 0 1 1 𝑣c 1 0 1 𝑣c 1 1 0 𝑣c
𝐿𝑀3 0 –1 1 −𝑣c –1 0 1 −𝑣c –1 1 0 −𝑣c
𝐿𝑀4 0 1 –1 −𝑣c 1 0 –1 −𝑣c 1 –1 0 −𝑣c
𝐿𝑅 6 0 –1 0 0 6 –1 0 –1 –1 7 0

Tab. A1. Parameters of voltage-controlled voltage generators for AND, OR, and XOR gates.

Parameter Value Parameter Value Parameter Value
𝑅on 10−2 𝑅off 1 𝑣c 1
𝛼 60 𝐶 10−9 𝑘 ∞
𝑉𝑡 0 𝛾 60 𝑞 10
𝑚0 –400 𝑚1 400 𝑖min 10−8

𝑖max 20 𝑘𝑖 10−7 𝑘𝑠 10−7

𝛿𝑠 0 𝛿𝑖 0

Tab. A2. Parameters of numerical simulations used in [1].

Appendix B: s-block

Fig. B1. Schematics of s-block. The s-block implements (7) and (9) according to the LTspice model in Listing C5.

Appendix C: SPICE Models

Listing C1. LTspice code for the self-organizing AND.

****Self-organizing AND ***************************
****Code for LTspice; tested with LTspice XVII *********
************************************************
.subckt SAND 1 2 3
.param zI=1E5 ; scaling factor
.param res={1*zI} vc=1
*DCM1
EM11 11 0 value={-V(2)+V(3)+vc}
Xmem11 1 11 memR
EM14 14 0 value={V(3)}
Xmem14 14 1 memR
EM13 13 0 value={4*V(1)+1*V(2)-3*V(3)-vc}
R11 13 1 {res}
*DCM2
EM21 21 0 value={-V(1)+V(3)+vc}
Xmem21 2 21 memR
EM24 24 0 value={V(3)}
Xmem24 24 2 memR
EM23 23 0 value={V(1)+4*V(2)-3*V(3)-vc}
R21 23 2 {res}
*DCM3
EM31 31 0 value={V(1)}
Xmem31 3 31 memR
EM32 32 0 value={V(2)}
Xmem32 3 32 memR
EM35 35 0 value={2*V(1)+2*V(2)-1*V(3)-2*vc}
Xmem35 35 3 memR
EM33 33 0 value={-4*V(1)-4*V(2)+7*V(3)+2*vc}
R31 33 3 {res}
*resistors
R1o 1 3 {res}
R2o 2 3 {res}
.ends SAND

Listing C2. LTspice code for the self-organizing XOR.

****Self-organizing XOR ****************************
****Code for LTspice; tested with LTspice XVII **********
*************************************************
.subckt SXOR 1 2 3
.param zI=1E5 ; scaling factor
.param res={1*zI} vc=1
*DCM1
EM11 11 0 value={-V(2)-V(3)+vc}
Xmem11 1 11 memR
EM12 12 0 value={V(2)+V(3)+vc}
Xmem12 1 12 memR
EM14 14 0 value={-V(2)+V(3)-vc}
Xmem14 14 1 memR
EM15 15 0 value={V(2)-V(3)-vc}
Xmem15 15 1 memR
EM13 13 0 value={6*V(1)-V(3)}
R11 13 1 {res}
*DCM2
EM21 21 0 value={-V(1)-V(3)+vc}
Xmem21 2 21 memR
EM22 22 0 value={V(1)+V(3)+vc}
Xmem22 2 22 memR
EM24 24 0 value={-V(1)+V(3)-vc}
Xmem24 24 2 memR
EM25 25 0 value={V(1)-V(3)-vc}
Xmem25 25 2 memR
EM23 23 0 value={6*V(2)-V(3)}
R21 23 2 {res}
*DCM3
EM31 31 0 value={-V(1)-V(2)+vc}
Xmem31 3 31 memR
EM32 32 0 value={V(1)+V(2)+vc}
Xmem32 3 32 memR
EM34 34 0 value={-V(1)+V(2)-vc}
Xmem34 34 3 memR
EM35 35 0 value={V(1)-V(2)-vc}
Xmem35 35 3 memR
EM33 33 0 value={-V(1)-V(2)+7*V(3)}
R31 33 3 {res}
*resistors
R1o 1 3 {res}
R2o 2 3 {res}
.ends SXOR

Listing C3. LTspice code for the self-organizing OR.

****Self-organizing OR *****************************
****Code for LTspice; tested with LTspice XVII *********
************************************************
.subckt SOR 1 2 3
.param zI=1E5 ; scaling factor
.param res={1*zI} vc=1
*DCM1
EM11 11 0 value={V(3)}
Xmem11 1 11 memR
EM14 14 0 value={-V(2)+V(3)-vc}
Xmem14 14 1 memR
EM13 13 0 value={4*V(1)+1*V(2)-3*V(3)+vc}
R11 13 1 {res}
*DCM2
EM21 21 0 value={V(3)}
Xmem21 2 21 memR
EM24 24 0 value={-V(1)+V(3)-vc}
Xmem24 24 2 memR
EM23 23 0 value={V(1)+4*V(2)-3*V(3)+vc}
R21 23 2 {res}
*DCM3
EM32 32 0 value={2*V(1)+2*V(2)-V(3)+2*vc}
Xmem32 3 32 memR
EM34 34 0 value={V(1)}
Xmem34 34 3 memR
EM35 35 0 value={V(2)}
Xmem35 35 3 memR
EM33 33 0 value={-4*V(1)-4*V(2)+7*V(3)-2*vc}
R31 33 3 {res}
*resistors
R1o 1 3 {res}
R2o 2 3 {res}
.ends SOR

Listing C4. LTspice code for the memristive element.

****Memristive element ************************************************
****Code for LTspice; tested with LTspice XVII *****************************
********************************************************************
.subckt memR plus minus
.param zI=1E5 zt=1E3 ; scaling factors
.param Ron={0.05*zI} Roff={1*zI} alpha={60*zI*zt} C={1E-9/zI/zt}
*model of memristive port
Gpm plus minus value={V(plus,minus)/(Ron+(Roff-Ron)*V(x))}
C1 plus minus {C} IC={0}
*integrator model
Gx 0 x value={-alpha*h(V(x),V(plus,minus))*V(plus,minus)/(Ron+(Roff-Ron)*V(x))}
Cx x 0 1 IC={mc(0.2,0.1)} ; randomized initial condition
Raux x 0 1G
*functions
.func h(x,vm)={u(x)*u(vm)+u(1-x)*u(-vm)}
.ends memR

Listing C5. LTspice code for VCDCG and s-block.

****Voltage-Controlled Differential Current Generator ************************
****Code for LTspice; tested with LTspice XVII *****************************
********************************************************************
.subckt VCDCG 1 2 3 4
*1: negative terminal (ground); 2: positive terminal;
*3: current output signal (voltage); 4: input from the s-circuit
.param zI=1E5 zt=1E3 ; scaling factors
.param m0={-400/zI*zt} m1={400/zI*zt} q={10/zI*zt} gamma={60*zt} cap={1E-3/zI/zt}
*model of VCDCG ports
G1 2 1 value={V(x)}
E1 3 1 value={V(x)}
R1 4 1 1G
C1 2 1 {cap} IC={0}
*integrator model
Gx 0 x value = {u(V(4,1)-0.5)*fdcg(V(2,1))-gamma*u(0.5-V(4,1))*V(x)}
Cx x 0 1 IC={0}
Raux x 0 1G
*functions
.param m0b={m0*pi/(2*q)} m1b={m1*pi/(2*q)}
.func fdcg(x)={q*(atan(m1b*(x+1))+atan(m0b*x)+atan(m1b*(x-1)))*2/pi}
.ends VCDCG

****8-input s-block ********************************************************************
****Code for LTspice; tested with LTspice XVII *********************************************
************************************************************************************
.subckt SCOMB8 1 2 3 4 5 6 7 8 9 10
*1: negative terminal (ground); 2: positive terminal (output); 3-10: inputs
.param zI=1E5 zt=1E3 ; scaling factors
.param imin={1E-8/zI} imax={10/zI} ks={2E3*zt} ki={2E3*zt}
*model
E1 2 1 value={V(s)}
R1 1 3 1G
R2 1 4 1G
R3 1 5 1G
R4 1 6 1G
R5 1 7 1G
R6 1 8 1G
R7 1 9 1G
R8 1 10 1G
*integrator model
Gs 0 s value ={-ks*V(s)*(V(s)-1)*(2*V(s)-1)-
+ki*(1-u(imin-abs(V(3,1)))*u(imin-abs(V(4,1)))*u(imin-abs(V(5,1)))*u(imin-abs(V(6,1)))*
+u(imin-abs(V(7,1)))*u(imin-abs(V(8,1)))*u(imin-abs(V(9,1)))*u(imin-abs(V(10,1)))-u(imax-abs(V(3,1)))*
+u(imax-abs(V(4,1)))*u(imax-abs(V(5,1)))*u(imax-abs(V(6,1)))*u(imax-abs(V(7,1)))*u(imax-abs(V(8,1)))*
+u(imax-abs(V(9,1)))*u(imax-abs(V(10,1))))}
Cs s 0 1 IC={0.75}
Raus s 0 1G
.ends SCOMB8


